

estimates store — Store and restore estimation results

[Description](#)
[Option](#)
[Also see](#)

[Quick start](#)
[Remarks and examples](#)

[Menu](#)
[Stored results](#)

[Syntax](#)
[References](#)

Description

`estimates store name` stores the current (active) estimation results under the name *name*.

`estimates restore name` loads the results stored under *name* into the current (active) estimation results.

`estimates query` tells you whether the current (active) estimates have been stored and, if so, the name.

`estimates dir` displays a list of the stored estimates.

`estimates drop namelist` drops the specified stored estimation results.

`estimates clear` drops all stored estimation results.

`estimates clear`, `estimates drop _all`, and `estimates drop *` do the same thing. `estimates drop` and `estimates clear` do not eliminate the current (active) estimation results.

Quick start

Store estimation results as `m1` for use later in the same session

```
estimates store m1
```

Restore estimation results from `m2`

```
estimates restore m2
```

Find out whether the current estimation results have been stored

```
estimates query
```

Display table of information about all stored results

```
estimates dir
```

Drop stored estimation results `m3`

```
estimates drop m3
```

Drop all stored results

```
estimates clear
```

Menu

Statistics > Postestimation

Syntax

```
estimates store name [, nocopy]
```

```
estimates restore name
```

```
estimates query
```

```
estimates dir [namelist]
```

```
estimates drop namelist
```

```
estimates clear
```

where *namelist* is a name, a list of names, `_all`, or `*`. `_all` and `*` mean the same thing.

Option

`nocopy`, used with `estimates store`, specifies that the current (active) estimation results are to be moved into *name* rather than copied. Typing

```
. estimates store hold, nocopy
```

is the same as typing

```
. estimates store hold  
. ereturn clear
```

except that the former is faster. The `nocopy` option is sometimes used by programmers.

Remarks and examples

`estimates store` stores estimation results in memory so that you can access them later.

```
. use http://www.stata-press.com/data/r15/auto
(1978 Automobile Data)
. regress mpg weight displ
(output omitted)
. estimates store myreg

. ... you do other things, including fitting other models ...

. estimates restore myreg

. regress
(same output shown again)
```

After `estimates restore myreg`, things are once again just as they were, estimationwise, just after you typed `regress mpg weight displ`.

`estimates store` stores results in memory. When you exit Stata, those stored results vanish. If you wish to make a permanent copy of your estimation results, see [R] [estimates save](#).

The purpose of making copies in memory is 1) so that you can quickly switch between them and 2) so that you can make tables comparing estimation results. Concerning the latter, see [R] [estimates table](#) and [R] [estimates stats](#).

Stored results

`estimates dir` stores the following in `r()`:

```
Macros
   r(names)      names of stored results
```

References

- Jann, B. 2005. Making regression tables from stored estimates. *Stata Journal* 5: 288–308.
- . 2007. Making regression tables simplified. *Stata Journal* 7: 227–244.

Also see

[R] [estimates](#) — Save and manipulate estimation results